# SNR for Standard and Weather Radar Equation

Marshall Bradley

*Algorithms for computing the Signal-to-Noise Ratio (SNR) achievable by idealized frequency modulated continuous wave radars optimized for atmospheric sounding are presented. Two cases are considered. In the first the target has reflectivity characterized by radar cross section* $\sigma$ *and echoes are proportional to* $\sigma r^{-4}$ *where r is slant range to the target. In the second the targets are distributed across volume with reflectivity* $\eta$ *per unit length. In this case echoes are proportional to* $\eta r^{-2}$*. In either case target detectability is limited by background noise characterized by the thermal noise spectral density* $T_{system} k_{Boltzmann}$ *where* $T_{system}$ *is the radar system noise temperature and* $k_{Boltzmann}$ *is Boltzmann's constant. Atmospheric absorption also limits detection range.*

## Background

Weather radar use a form of the radar equation that is different from the radar equation used in other disciplines. Consider a radar with transmit power $P_t$ and transmit antenna gain $G_t$. The transmit power $P_t$ is measured in watts (W) and the antenna gain is a measure of the ability of the radar to concentrate the transmit power in a particular direction. The power density $S_t$ of the transmit signal at distance $R$ from the radar is

$$S_t = \left((P_t G_t)/\left(4 \pi R^2\right)\right).$$

An omnidirectional transmitter has an antenna gain of unity. Practical radars often used transmit antennas with high gains in order to offset signal losses due to two-way propagation from the radar to the target and back. From a modeling standpoint it is convenient to think about the radiation characteristics of antennas in terms of current distribution over an aperture. If electrical current with wavelength $\lambda$ is induced to flow over a circular planar aperture with radius $a$, then amplitude scale directivity (proportional to radiated electric field strength) of the antenna $D(\theta)$ can be shown to be

$$D(\theta) = \frac{2 J_1(k a \sin \theta)}{k a \sin \theta}$$

where $\theta$ is polar angle measured with respect to the normal to the face of the antenna and $k = 2 \pi / \lambda$ is wavenumber. The gain of this antenna is defined to be

$$G_t = \frac{4 \pi}{2 \pi \int_0^{\pi} D^2(\theta) \sin \theta \, d\theta},$$

which is just the ratio of the area of a unit sphere to the area projected on the sphere by the transmit beam pattern. Antenna gain (for either the transmit or receive antennas) can also be expressed in terms of the ratio

$G = \frac{4\pi}{\Omega}$,

where Ω is the solid angle of the radar beam in steradians. If we envision radiation being constant within the solid angle Ω and zero outiside this angle, then

$\Omega = \int_0^{2\pi} \int_0^{\theta_r} \sin\theta \, d\theta \, d\phi = \frac{\pi}{4}\,\theta_r^2$

where $\theta$ is polar angle in spherical coordinates, $\phi$ is the spherical coordinate azimuthal angle and $\theta_r$ is the radar half-beam width in radians.

Now consider a target located at distance $R$ from the radar with radar cross section $\sigma$. If the target were a large sphere of radius $r_0 >> \lambda$, then the $\sigma = \pi\, r_0^2$. The power captured by target is

$P_c = \sigma\, S_t$.

This power is radiated back towards the radar. The power per unit area of the backscattered radiation at the radar location is

$$S_r = \frac{P_c}{4\,\pi\,R^2} = \frac{\sigma\,P_t\,G_t}{\left(4\,\pi\,R^2\right)^2}.$$

The backscattered power received by the radar is

$$P_r = S_r\,A_r = \frac{\sigma\,P_t\,G_t\,A_r}{\left(4\,\pi\,R^2\right)^2}$$

where $A_r$ is the effective area of the receiver antenna. If $G_r$ denotes the gain of the receiver antenna, then

$$A_r = \frac{\lambda^2\,G_r}{4\,\pi}$$

where $\lambda$ is the wavelength of the radar carrier frequency. In terms of antenna gain the radar equation (received power) can be written

$$P_r = P_t\,\frac{\sigma\,G_t\,G_r\,\lambda^2}{(4\,\pi)^3\,R^4}.$$

In weather radar applications, scattering occurs from small particles, typically drops of water, or insects that are distributed across a volume. This is known as volume scattering and the appropriate representation of the radar cross section is

$\sigma = \eta\,V$

where $\eta$ is the radar cross section per unt volume ($\mathrm{m}^{-1}$) and $V$ is the illuminated volume. At distance $R$ from the radar this volume is

$V = \Omega_t\,R^2\,\frac{c}{2B}$

where $\Omega_t$ is the solid angle of the radar transmit beam, $c$ is the speed of light and $B$ is the bandwidth of the radar transmission. The factor

$$dr = \frac{c}{2B}$$

is the range resolution of the radar.

In order to compute the radar cross section for a volume distribution of small particles, we first consider the radar cross section of a small water particle. Scattering from such an object is known as Rayleigh scattering and the result is

$$\sigma_{\text{drop}} = \frac{\pi^5}{\lambda^4} \left| K \right|^2 D^6, \quad D < \lambda/16$$

where $D$ is the drop diameter (m) and $|K|^2 = 0.93$ for water. If there were $N$ particles distributed across the volume $V$, then the radar across section would be

$$\sigma = N\, \sigma_{\text{drop}}.$$

This leads to the following representation for $\eta$, the radar cross section per unit volume,

$$\eta = \frac{\pi^5}{\lambda^4} \left| K \right|^2 Z_m$$

where $Z_m$ is the reflectivity factor ($m^3$). Meteorologists express the reflectivity factor in units of (millimeters)$^6$/(meters)$^3$. This leads to

$$Z_m = Z \times 10^{-18}.$$

The radar cross section per unit volume in terms of $Z$ is

$$\eta = 10^{-18} \frac{\pi^5}{\lambda^4} \left| K \right|^2 Z.$$

The reflectivity factor $Z$ is always expressed on a decibel scale and referred to as $dBZ$. In terms of $dBZ$, the radar cross section per unit volume is

$$\eta = 10^{-18} \frac{\pi^5}{\lambda^4} \left| K \right|^2 10^{\frac{dBZ}{10}}.$$

The radar equation can be written as

$$P_r = P_t \frac{G_t\, G_r\, \lambda^2}{(4\pi)^3 R^4} \left(\Omega R^2 \frac{c}{2B}\right)\left(10^{-18} \frac{\pi^5}{\lambda^4} \left| K \right|^2 10^{\frac{dBZ}{10}}\right)$$

or equivalently as

$$P_r = P_t \frac{G_r}{(4\pi R)^2 \lambda^2} \left(\frac{c}{2B}\right)\left(10^{-18} \pi^5 \left| K \right|^2 10^{\frac{dBZ}{10}}\right).$$

Ultimately radar performance is limited by thermal noise. The thermal noise spectral density (W/Hz) of a radar system temperature $T_s$ is

$$N_0 = T_{\text{system}}\, k_{\text{Boltzmann}}$$

where the system noise temperature $T_{\text{system}}$ is measured in degrees Kelvin and $k_{\text{Boltzmann}}$ is Boltzmann's constant

$$k_{\text{Boltzmann}} = 1.38 \times 10^{-23}\ \text{W/deg K/Hz}.$$

FMCW radars use multiple pulse processing in order to increase the radar detection range and sensitivity to variations in target velocity.  If $N_{\text{stack}}$ pulses each of length $T_m$ are used in the radar processing architecture, then the radar coherent integration time is

$T_{\text{int}} = N_{\text{stack}}\, T_m$.

The signal to noise ratio *SNR* for the weather radar system against a volumetric, weather target at range *R* with reflectivity *dBZ* is

$$SNR_{\text{weather}} = P_t \frac{\frac{G_r}{(4\pi R)^2 \lambda^2}\left(\frac{c}{2B}\right)\left(10^{-18}\pi^5 \left|K\right|^2 10^{\frac{dBZ}{10}}\right)}{\frac{1}{N_{\text{stack}}\, T_m} T_{\text{system}}\, k_{\text{Boltzmann}}}.$$

The signal to noise ratio *SNR* for the same radar system against an ordinary target with radar cross section $\sigma$ is

$$SNR_{\text{standard}} = P_t \frac{\frac{\sigma G_t G_r \lambda^2}{(4\pi)^3 R^4}}{\frac{1}{N_{\text{stack}}\, T_m} T_{\text{system}}\, k_{\text{Boltzmann}}}.$$

# Algorithms

In[ ]:=

## Radar absorption in the troposphere

The algorithm $\gamma$fh[f,h] defined in the following block of code computes atmospheric absorption in units of dB/km over the frequency range 0.1 GHz to 100 GHz and over the altitude range 0 to 30,000 meters. The results here are based upon the work of Blake (1991). The two molecules that absorb radar energy in the 0.1 GHz to 100 GHz range are water vapor in gaseous form and oxygen.  Water vapor is resonant at 22 GHz. Oxygen is resonant at 60 GHz, but there are actually closely spaced resonances over the 50 GHz to 70 GHz band. Of the two, oxygen is the greater absorber. There is an additional oxygen resonance at 120 GHz but its effect is not significant at frequencies below 100 GHz.

The underlying computations supporting the algorithm have been precomputed and are coded in image format in order to enhance speed of computation:

In[ ]:=

```
Module[{γImage, data, data1, γhf},
 γImage =
```

```
;
 data = ImageData[γImage];
 data1 = Table[{(row - 1) 500, 0.1 + (col - 1) 0.1, data⟦row, col⟧},
   {row, 1, 95}, {col, 1, 1000}];
γhf = Interpolation[ Flatten[data1, 1], Method → "Spline"];
 AbsorptionTropospheric[f_, h_] := γhf[h, f]]
```

Absorptions at a wide range of frequencies and altitudes are shown in the following plot:

In[◦]:=
```
LogLogPlot[{AbsorptionTropospheric[f, 0], AbsorptionTropospheric[f, 5000],
  AbsorptionTropospheric[f, 10 000], AbsorptionTropospheric[f, 15 000],
  AbsorptionTropospheric[f, 20 000], AbsorptionTropospheric[f, 25 000],
  AbsorptionTropospheric[f, 30 000]}, {f, 0.1, 100}, ⋯]
```

Out[◦]=




In[◦]:=

## Weather radar equation

An algorithm for making weather radar signal-to- noise (SNR) computations is implemented here. Input units are range *r0* (m), *dBZ* (decibel), *SystemNoiseTemperature* (°K), *TxPower* (Watts), *antennagain* (decibel),carrier frequency *fc* (Hz), pulse length *Tm* (sec), sweep width *B* (Hz), stack size *Nstack* (positive integer):

In[ ]:=
```
SNRWeatherRadar[r0_, dBZ_, SystemNoiseTemperature_, TxPower_,
  antennagain_, fc_, Tm_, B_, Nstack_] := Module[{c, λ, kBoltzmann,
   ThermalNoise0, Ω, K, η, Gt, Gr, radarconstant, fGHz, αdbPerM, EchoPower},
  c = 3.0 * 10^8; λ = c / fc;
  kBoltzmann = 1.38 * 10^-23;
  ThermalNoise0 = kBoltzmann * SystemNoiseTemperature;
```

$$\Omega = \frac{4\pi}{10^{\frac{antennagain}{10}}};\ K = (0.93)^{1/2};\ \eta = 10^{-18}\frac{K^2 \pi^5 10^{\frac{dBZ}{10}}}{\lambda^4};$$

$$Gt = 10^{antennagain/10};\ Gr = Gt;$$

$$radarconstant = \frac{Gt * Gr * \lambda^2}{(4\pi)^3}\Omega\frac{c}{2B}\eta;\ fGHz = fc / 10^{9.0};$$

```
  αdbPerM = AbsorptionTropospheric[fGHz, 0] / 1000;
```

$$EchoPower = TxPower\frac{radarconstant * 10^{-\alpha dbPerM*2\, r0/10}}{r0^2};$$

$$10\,Log\left[10, \frac{Nstack * Tm * EchoPower}{ThermalNoise0}\right]]$$

Sample computation:

In[ ]:=
```
r0 = 1000.0;
dBZ = -30;
SystemNoiseTemperature = 60.0;
TxPower = 3.0;
antennagain = 30;
fc = 33.4 * 10^9;
Tm = 190.0 * 10^-6;
B = 48.0 * 10^6; Nstack = 256;
SNRWeatherRadar[r0, dBZ, SystemNoiseTemperature,
 TxPower, antennagain, fc, Tm, B, Nstack]
```

Out[ ]=

10.7204

In[ ]:=

## Standard radar equation

An algorithm for making standard radar signal-to- noise (SNR) computations is implemented here. Input units are range *r0* (m), radar cross section $\sigma$(square meter), *SystemNoiseTemperature* (°K), *TxPower* (Watts), *antennagain* (decibel),carrier frequency *fc* (Hz), pulse length *Tm* (sec), sweep width *B* (Hz), stack size *Nstack* (positive integer):

In[ ]:=
```
SNRStandardRadar[r0_, σ_, SystemNoiseTemperature_,
   TxPower_, antennagain_, fc_, Tm_, B_, Nstack_] :=
  Module[{c, λ, kBoltzmann, ThermalNoise0, Gt, Gr, fGHz, αdbPerM, EchoPower},
   c = 3.0 * 10^8; λ = c / fc;
   kBoltzmann = 1.38 * 10^-23;
   ThermalNoise0 = kBoltzmann * SystemNoiseTemperature;

   Gt = 10^(antennagain/10);
   Gr = Gt;
   fGHz = fc / 10^9.0;
   αdbPerM = AbsorptionTropospheric[fGHz, 0] / 1000;
   EchoPower = TxPower (Gt * Gr * λ^2)/((4 π)^3 r0^4) σ * 10^(-αdbPerM*2 r0/10);
   10 Log[10, (Nstack * Tm * EchoPower)/ThermalNoise0]]
```

Sample computation:

In[ ]:=
```
r0 = 1000.0;
σ = 10^-5.0;
SystemNoiseTemperature = 60.0;
TxPower = 3.0;
antennagain = 30;
fc = 33.4 * 10^9;
Tm = 190.0 * 10^-6;
B = 48.0 * 10^6; Nstack = 256;
SNRStandardRadar[r0, σ, SystemNoiseTemperature,
 TxPower, antennagain, fc, Tm, B, Nstack]
```

Out[ ]=
```
18.3725
```

A comparison is shown in the following:

In[◦]:=
```
dBZ = -30;
σ = 10^-5.0;
SystemNoiseTemperature = 60.0;
TxPower = 3.0;
antennagain = 30;
fc = 33.4 * 10^9;
Tm = 190.0 * 10^-6;
B = 48.0 * 10^6; Nstack = 256;
LogLinearPlot[
 {SNRWeatherRadar[r, dBZ, SystemNoiseTemperature, TxPower, antennagain, fc,
   Tm, B, Nstack], SNRStandardRadar[r, σ, SystemNoiseTemperature, TxPower,
   antennagain, fc, Tm, B, Nstack]}, {r, 100, 5000}, ⋯]
```

Out[◦]=




# Data analysis

We begin with some data recorded by a WiPPR prototype on 30 January 2013 during a period of light snow at a location near Dugway Utah. The data shown in the images below are stored in an SNR format on a decibel scale. Beams 1-4 respectively point in the north, east, south and west directions. All beams make an angle of 10 deg with respect to the vertical:

In[◦]:= images = { , , ,  };

Lets begin by looking at the SNR data in detail:

In[ ]:=
```
lbls = {"1: north", "2: east", "3: south", "4: west"};
g = Table[
   ReliefPlot[ImageData[images⟦i⟧], ⋯ ], {i, 1, 4}];
GraphicsGrid[{{g⟦1⟧, g⟦2⟧}, {g⟦3⟧, g⟦4⟧}}]
```

Out[ ]=




Look at a slice of data from the first image at the 100th range:

In[ ]:=
```
i = 1; nr = 100;
snr = ImageData[images〚i〛];
ListLinePlot[snr〚nr〛, ...]
```

Out[ ]=




Select and plot all data with SNR greater than 2 dB:

In[ ]:=
```
snrvsrange = Table[
   snr = ImageData[images〚i〛];
   Select[Table[{3.125 nr, Max[snr〚nr〛]}, {nr, 11, 512}], #〚2〛 > 2.0 &], {i, 1, 4}];
ListPlot[snrvsrange, ...]
```

Out[ ]=




Fit the data to a curve of the form $a + b\log(_{10} r)$ where $r$ is slant range. If $b = -20$ then the two way spreading is spherical:

In[ ]:=
```
data = Flatten[snrvsrange, 1];
Clear[a, b];
model = a + b * Log[10, r];
fit = FindFit[data, model, {a, b}, r]
```

Out[ ]=

```
{a → 57.9693, b → -16.1695}
```

The spreading is almost spherical.

Compare the fit to the data:

In[ ]:=
```
{a, b} = {a, b} /. fit;
{rmin, rmax} = MinMax@Map[First, data];
epi = Line[Table[{r, a + b * Log[10, r]}, {r, rmin, rmax, 5}]];
ListPlot[snrvsrange, Epilog → epi, ⋯]
```

Out[ ]=




# References

Blake, Lamont V. (1991), *Radar Range - Performance Analysis*, Munro Publishing Co.